

Intusoft Newsletter

Personal Computer Circuit Design Tools

January 1993 Issue



Copyright © Intusoft, All Rights Reserved
intusoft (310) 833-0710 Fax (310) 833-9658


Simulating Fuzzy Logic

Control systems are an integral part of many electronic devices. Simulation of the control mechanism along with the input and output circuitry can be quite valuable when evaluating different design concepts and specifications. IsSPICE is capable of simulating both the control mechanism and the circuitry to be controlled at the same time. This allows the total system to be simulated. In this article, we will explore simulation of two control systems. One using a traditional PID approach, plus a second newly emerging technique, Fuzzy Logic.

Although still in its infancy, fuzzy technology offers tremendous potential for enhancing products we use every day. Fuzzy logic has been applied to machine automation, industrial testing, and most recently to household appliances. In the very near future, everything from washers to heating systems will be fuzzy logic controlled with benefits like improved efficiency and superior performance. Currently, camcorders are correcting user induced camera jiggle and focusing anomalies with fuzzy logic.

In This Issue

- 2 **Simulating Fuzzy Logic**
- 6 Continuous PID Control
- 8 Simulating Mixed Mode Circuits
- 12 The Modeling Corner:
New Vendor Models
Open More Applications



Continued on Page 28-2

Simulating Mixed Mode Circuits

In order to effectively handle the wide variety of simulation requirements, models for digital functions must be available in several forms. Transistor level representations using the exact topology of the IC are very important for studying transmission line loading and driving effects. Without the transistor level models, very accurate studies are virtually impossible. While modeling large ICs at this level is impractical, MSI circuits can easily be constructed. Gate level primitives must be available to allow assembly of combinational logic and larger ICs. Finally, a behavioral modeling approach is required in order to allow large complex devices to be simulated rapidly. Intusoft provides models using all of these levels. In this article, we will explore the method used to model primitive gate elements and how to perform some basic mixed mode simulations.

Continued on Page 28-8

28-1

Simulating Fuzzy Logic Circuits


continued
from Page
28-1


Unlike binary systems, such as the Boolean logic simulations explored in the Mixed Mode article on page 28-8, fuzzy control uses comparative models, or rules, to determine a response to input analog stimuli. These rules enable fuzzy logic to approximate the nature of human reasoning and provide a precise and accurate control strategy. The name “fuzzy” comes from “fuzzy sets”, the basis of fuzzy logic, originally developed by Professor Lotfi Zadeh of U.C. Berkeley in 1965. References 1, 2 and 3 contain introductory material on this topic.

The IS_{SPICE3} program contains all of the tools needed to simulate fuzzy logic systems. Using the example shown in Figure 1 [ref. 2], we will simulate an open loop control system with temperature and humidity as the input quantities and fan speed as the output.

A Fuzzy Fan Controller

The first processing step in a fuzzy control system is the conversion of a set of input values, called fuzzy variables, into a set of output levels, called membership values. The transfer function relationship between the fuzzy variables and the membership values is called a Membership Function; for example, the trapezoidal response between the input Humidity and the output





function labeled Moist in Figure 2. Voltage is used as the analog of temperature in degrees and humidity in % to represent the fuzzy variables. The membership functions are implemented with the new ISPIECE3 “If-Then-Else” statement and schematically represented as the series of B elements (B1-B6) in Figure 1. The If-Then-Else function is implemented using the ISPIECE3 nonlinear dependent source:

Bname Out+ Out- V = Logical_Expression ? C : D

This reads as “If the Logical_Expression is true Then C, Else D”. The logical expression, C, or D may be a constant or an expression containing any algebraic equation made of various mathematical operators, node voltages and currents. The output may be either voltage or current and the logical expression is usually $A < B$ or $A > B$, where A and B can be expressions or constants. Note that voltages and currents within the simulation may be analogs of any quantity, i.e. temperature, speed.

Referring to Figure 2, the syntax describing the cool and warm functions is as follows:

```
*COOL FUZZY INPUT MEMBERSHIP
B1 1 0 V= V(3) < 60 ? 1 : V(3) < 75 ? 5 - V(3) / 15 : 0
*WARM FUZZY INPUT MEMBERSHIP
B2 20 0 V= V(3) < 75 ? -4 + V(3) / 15 : V(3) < 90 ? 6 - V(3) / 15 : 0
BLIMIT 2 0 V= V(20) < 0 ? 0 : V(20)
```

Rule #	Condition	Conclusion
1	If temperature is COOL AND humidity is DRY	Then Fan-Speed is Off
2	If temperature is COOL AND humidity is MOIST	Then Fan-Speed is On_Low
3	If temperature is COOL AND humidity is WET	Then Fan-Speed is On_Low
4	If temperature is WARM AND humidity is DRY	Then Fan-Speed is On_Low
5	If temperature is WARM AND humidity is MOIST	Then Fan-Speed is On_High
6	If temperature is WARM AND humidity is WET	Then Fan-Speed is On_High
7	If temperature is HOT	Then Fan-Speed is On_High

Figure 3, Fuzzy Rules. The AND condition is processed using a logical product (minimum of the two values). Conclusions controlling the same Fan-Speed output state are processed using a logical sum (maximum value).

The other functions are shown in the netlist in Table 1. The membership functions can take on any value between 0 and 1. The overlap of the functions allow a continuous translation between the input and output quantities. The flexible analog behavioral element, B, allows virtually any membership function to be entered (Bell, triangle, trapezoid).


Processing The Membership Functions

The second step towards generating an output signal is to process the membership values using a set of fuzzy rules (Figure 3). The rules are composed of a condition (antecedent section) and a conclusion (consequent section); both of which are handled by the B element. In the case where there is more than one value used in the condition (AND clause), the logical product (minimum) must be determined. This is the case for all but rule 7. For example, if the COOL value were .6 (Temperature \cong 65°) and the DRY value was .3 (Humidity \cong 28%), then the rule 1 output would be .3. The MIN subcircuit is used for this purpose. Notice that the rules are processed in parallel. This allows fuzzy logic to control a complex system using simple expressions resulting in very high processing speed and excellent error tolerance.

A logical sum (maximum) is used to resolve the rules that compete for control of the same output state; for instance, the fan-speed On_Low condition. For example, the MAX3 subcircuit (Table 1) will take the maximum membership value from the output of rules 2, 3 and 4.

Defuzzification; The Final Step

In order to produce a “crisp” output value, the final results from the processed membership values must be unified. In this example,



```

FUZZY LOGIC CONTROL SYSTEM
.TRAN 01 1
.PRINT TRAN V(3) V(4) V(14) V(15) V(16) V(17)
V1 3 0 PWL 0 100 1 50
V2 3 0 PWL 0 0 1 100
*COOL FUZZY INPUT MEMBERSHIP
B1 1 0 V= V(3) < 60 ? 1 : V(3) < 75 ? 5 - V(3) / 15 : 0
*WARM FUZZY INPUT MEMBERSHIP
B2 20 0 V= V(3) < 75 ? -4 + V(3) / 15 : V(3) < 90 ? 6 - V(3) / 15 : 0
BLIMIT 2 0 V= V(20) < 0 ? 0 : V(20)
*HOT FUZZY INPUT MEMBERSHIP
B3 5 0 V= V(3) < 75 ? 0 : V(3) < 90 ? -5 + V(3) / 15 : 1
*DRY FUZZY INPUT MEMBERSHIP
B4 6 0 V= V(4) < 20 ? 1 : V(4) < 40 ? 2 - V(4) / 20 : 0
*MOIST FUZZY INPUT MEMBERSHIP
B6 80 0 V= V(4) < 40 ? -1 + V(4) / 20 : V(4) > 60 ? 4 - V(4) / 20 : 1
BLIMIT2 8 0 V= V(80) < 0 ? 0 : V(80)
*WET FUZZY INPUT MEMBERSHIP
B5 7 0 V= V(4) < 60 ? 0 : V(4) < 80 ? -3 + V(4) / 20 : 1
* FUZZY RULES
XRULE1 1 6 14 MIN
XRULE2 1 8 9 MIN
XRULE3 1 7 10 MIN
XRULE4 2 6 11 MIN
XRULE5 2 8 13 MIN
XRULE6 2 7 12 MIN
X7 13 12 7 16 MAX3
X14 9 10 11 15 MAX3
* DEFUZZIFICATION
BAVERAGE 17 0 V=(V(15)*40) + (V(16)*100) / (V(14) + V(15) + V(16))

```

Table 1, The IsSPICE3 netlist for the Fuzzy Logic Fan-Speed Control System. B elements that use other nodes as input can be placed inside subcircuits in order to avoid node renumbering problems.

```

*INCLUDE FUZZY.LIB
.SUBCKT MIN 1 2 3
*Connections In1 In2 Out
BMIN 3 0 V=V(1) < V(2) ? V(1) : V(2)
.ENDS
.SUBCKT MAX3 1 2 3 4
*Connections In1 In2 In3 Out
BMAX1 5 0 V=V(1) > V(2) ? V(1) : V(2)
BMAX2 4 0 V=V(3) > V(5) ? V(3) : V(5)
.ENDS
.END

```

the two outputs of the logical sum (Max3 subcircuits) blocks and rule 1 are reduced to a single value by taking a weighted average. Other defuzzification methods are possible.

The circuit in Figure 1 was tested by sweeping the voltages corresponding to the temperature, V(3), and humidity, V(4), in order to verify that the correct membership functions were produced. Then temperature was swept from 50 to 100 to 0 degrees and humidity from 100 to 0 to 50% to produce the output shown in Figure 4. The total simulation time is negligible, less than 4.3s, revealing that vastly more complex systems can be easily handled.

Schematics And The B Element

The B element may contain expressions using algebraic equations, If-Then-Else notation, or Boolean logic sequences. The expressions may include references to other nodes (V(3)) or voltage sources (I(V1)). In a schematic environment, this “text” based connection can cause a number of problems if the nodes in the schematic are reordered. In order to avoid this renetting problem, SPICENET includes a continuation symbol. Once this symbol is placed on a node its number will remain unchanged. B elements may safely use a nodal reference after a continuation has been placed on the node. However, care should be taken to review any references in the B element syntax before simulation or if IsSPICE3 states that a syntax error has been made.


Continuous PID Control System [4]

When severe requirements are placed on the precision of an automatic control process, then proportional controllers, which amplify the difference between the desired and measured quantities, are inadequate. In complex plants, there are often many phase lags which change the negative feedback into positive feedback at higher frequencies. When the total phase lag of the open loop reaches 180° , before the gain drops below one, the closed loop will become unstable. This instability may be circumvented by reducing the proportional gain of the controller. However, the step response of the output will no longer be satisfactory. As a remedy an additional signal corresponding to the derivative of the control deviation can be added to the proportional quantity. This provides adequate differential gain which leads to an improved dynamic response in the control process.

In addition to the transient behavior, the steady state deviation is a valid criteria for the quality of a control process. External disturbances will cause unbalancing forces, but proportional gain by itself is not adequate enough to keep the control deviation at zero. For this reason, the deviation can be integrated and added to the proportional and differential quantities. In this way, even the smallest deviations are summed up continuously and produce an actuator response great enough to adjust the steady state deviation to zero.

A system using these quantities for control is called a PID control system (**P**roportional, **I**ntegral, and **D**erivative) and can easily be simulated using IS_{SPICE}'s Laplace and Z transform capabilities. PID control is used in situations where a process can be very precisely described using a mathematical model. In real world systems, creating such accurate models can be difficult. This is where Fuzzy Logic control can become a valuable alternative.

The diagram in Figure 5 shows a simple 3 pole system. The difference between the input and the output quantities is fed into gain, integral and differential blocks. The P, I, and D results are






Figure 6, The circuit simulates the PID control system described in Figure 5. The AC and transient analysis results using Laplace expressions are shown above. Simulation using Z transform elements is also possible [4].

then summed and filtered through an s-domain expression; in this case a series of three single pole filters. The gain and pole frequencies are entered through the schematic and passed into the generic Laplace subcircuits.

IsSPICE3 Makes A Fuzzy Connection

While not intended as a replacement for fuzzy logic development tools, IsSPICE3 provides some significant advantages. IsSPICE3 simulations are not limited to the fuzzy logic problem and its implementation. The fuzzy logic, and its inputs and outputs, can be simulated right along with the circuitry they are used to control. This is especially important when the circuit response can affect the design criteria of the control system.

[1] Omron Electronics, Inc., "An Introduction to Fuzzy Logic and its Application in Control Systems", 1991, 1-800-82-OMRON

[2] Ken Anderson, "Control Systems Sample Life in the Fuzzy Lane", Personal Engineering and Instrumentation News, 10/92, 1-603-427-1427

[3] Greg Viot, "Fuzzy Logic in C", Dr. Dobb's Journal, Feb. 1993, pg. 40, 1-303-447-9330

[4] Karl H. Muller, "A SPICE Cookbook", Intusoft, 1991


Simulating Mixed Mode Circuits

continued
from Page
28-1

IsSPICE3 is a native mixed mode simulator with both the digital and analog elements simulated at the same time using continuous voltage and current levels. This allows one schematic with one concise netlist to describe the entire circuit. No special interfacing between the analog and digital elements is necessary and both types of components may be freely mixed. The complexity of the driving/loading characteristic in the digital elements is hidden in the subcircuits and may be altered by the user to the desired state.

The modeling of digital functions at the behavioral and gate level is handled by the nonlinear dependent source (B) in IsSPICE3. The B element allows all types of Boolean logic expressions to be entered. The operators are ~ - Not, & - And, and | - Or. For example, B1 3 0 V=V(1) & V(2) would And the voltages at nodes 1 and 2. Much more complex Boolean expressions can be used. Logic primitives can be easily created using simple Boolean expressions such as the NAND gate example shown in Figure 7. The RC combination on the output of the NAND3 subcircuit is used to add propagation delay and rise/fall time. A library, enclosed on the Newsletter floppy disk, includes a set of gates, flip-flops, Schmitt triggers, and latches.

There are three new .OPTIONS variables available to control the global Boolean logic levels. They are LONE (logic 1), LZERO (logic 0), and LTHRESH (threshold marking the logic 0 and 1 transition). However, virtually all logic families and any set of logic levels can be mixed during the same simulation.



```

Models used in Figures 7 and 8
.SUBCKT DFLOP 1 2 3 4 5 6
*CLRN D CLK PREN Q QN
X1 4 12 11 13 NAND3 {IC=0}
X2 13 1 3 11 NAND3 {IC=1}
X3 11 3 12 10 NAND3 {IC=0}
X4 10 1 2 12 NAND3 {IC=1}
X5 4 11 6 5 NAND3 {IC=0}
X6 5 1 10 6 NAND3 {IC=1}
.ENDS
.SUBCKT NAND3 1 2 3 4 {IC=0}
B1 44 0 v=-(v(1)&v(2)&v(3))
R1 44 4 1
C1 4 0 .87NF IC={IC}
.ENDS
.SUBCKT MSCHM 1 2
*Connections VControl Vout
V1 3 0 3.4
R1 3 4 1K
S1 4 0 1 0 SWMOD
.MODEL SWMOD SW(ROn=10
+ ROFF=100K VT=1.8 VH=-.2)
B1 5 0 V=V(4)
R2 5 2 10
C1 2 0 100P
.ENDS

```


Digital Phase Comparator

One of the most widely used circuits containing mixed analog and digital elements is the Phase Lock Loop. A major component in the PLL is the phase comparator. Figure 7 shows the schematic for the MC4344 digital phase comparator. It accepts TTL input waveforms and produces error signals that are proportional to the phase and/or frequency difference of the input signals.


Initialization of the comparator is mandatory and determines whether the R or V input is viewed as being the leading signal. Figure 8 shows the results of the simulation of both phase detectors in the MC4344 for out of phase V/R inputs.

Ring Oscillator Using Boolean Logic

The circuit in Figure 9 illustrates the use of a D flip-flop with preset and clear in a mixed mode circuit sent to us by Mike Penberth of Technology Sources, the Intusoft representative in the U.K. Note that oscillation is started by using the UIC keyword and initial condition on the capacitor. The Schmitt trigger element is listed in the MSCHM subcircuit in Figure 7.

Initialization And Time Step Control

While digital simulators may run with unknown output states, this condition is not allowed in SPICE. Therefore, proper initialization






Figure 9, A ring oscillator built using Boolean logic gates. The customized Schmitt trigger element (MSCHM) is easily created with the IsSPICE3 built-in switch model (See Figure 7 for listing).

of circuits containing memory or feedback is mandatory. Circuits without feedback normally converge without any assistance. For example, the flip-flops in the ring oscillator are initialized using the preset line. In the MC4344, the gates are initialized using the .IC commands. If the initial condition commands, .IC and IC= are used to predefine the circuit node voltages, the UIC directive in the .TRAN statement must also be invoked.

It should be noted that the Boolean logic expressions do not have any capacitive delays and do not contribute to the control of the time step. Therefore, it may be necessary to force the time step to a specific limit using the .TRAN TMAX parameter. In the ring oscillator circuit two simulations were run; one without a TMAX specification and another with $TMAX = TSTEP$. In this case there was little variation in the flip-flop output. The simulation time went from 5.83s to 15.32s. In the case of the digital phase comparator, the circuit failed to function properly unless TMAX was set to at least 1/5th of TSTEP. Control of the time step can be crucial to getting accurate data and even to the circuit functioning properly.

SPICE transient simulations should be approached with a general caution. Allowing SPICE to control the time step, especially with small circuits (< 100 components) can cause data to be aliased even in SPICE programs that output the individual internal time steps. If you do not see the correct waveform shapes you think you should be getting, make TMAX 1/2 to 1/10th of the TSTEP value. Another rule of thumb, if only a small number of data points are taken, is to make the TMAX value 1/2 to 1/10th of the input stimulus waveforms' rise/fall time (V(6) & V(9) in Fig. 7).

Modeling Corner

In this issue of The Modeling Corner we will bring you up to date on some of the newly released models from several hardware vendors. The models, available directly from the vendors or Intusoft, are designed to work with virtually all SPICE programs including IsSPICE.

Analog Devices, Burr-Brown, Elantec, Harris, and Linear Technology have all put out a series of linear IC models. To this list of vendors we can now add new model libraries from Apex Microtechnology, makers of a complete line of power op-amps (Tucson Az., 602-742-8600), and Zetex, who offers over 230 "European part" models for BJTs, Darlingtons, and Mosfets. Zetex is based in the U.K (061-627-5105) with U.S. offices in New York (516-543-7100). Shown below is a list of the new model additions from each of the above companies.

Analog Devices: Instrumentation amps, op-amps; AD620, AD620A, AD620B, AD620S, AD797, AD797A, AD797B, AD797S, AD9621, AD9622, AD9623, AD9624, OP213, OP295, SSM-2017, SSM-2017P

Apex: power, power booster, and wideband amplifiers up to 30A output current and 300V supply voltage; PA01, PA02, PA03, PA04, PA05, PA07, PA08, PA09, PA10, PA12, PA19, PA21, PA25, PA30, PA41, PA51, PA81, PA83, PA84, PA85, PA88, PA89, PB50, PB58, PB58A, WA01, WA05

Burr-Brown: Multiplexor, current feedback op-amps, precision op-amps, switched integrator, wideband switched input op-amp; MPC100X1, OPA603, OPA604, OPA604E, OPA77, OPA77E, ACF2101M, OPA671M, OPA675M, OPA676M

Comlinear: current feedback op-amps; CL400, CL401, CL404, CL406, CL409, CL410, CL414, CL415, CL420, CL430, CL501, CL505

Elantec: high speed op-amps, video distribution amps; EL2044C, EL2071, EL2073, EL2074, EL2075, EL2099

Linear Technology: op-amps; LT1122, LT1220, LM10C

Zetex: 80 NPN BJTs and 70 PNP BJTs including 2N, 2SC, BC, BCX, BF, BFQ, FCX, FMM, FZT, MPS, and ZTX parts; 57 Diodes including BAL, BAR, BAS, BAV, BBY, FFMD, FMMV, HD, LD, ZC, and ZDX parts; 10 Darlington BJTs and 15 Mosfets

All of the models mentioned in The Modeling Corner are available to Intusoft Newsletter subscribers on the enclosed Newsletter floppy disk. Other interested parties may purchase the floppy disk associated with this newsletter for a nominal fee. Over 940 other vendor supplied models are also available.

Power Op-amps From Apex

The IsSPICE subcircuits from Apex model high power op-amps. These models enable applications like audio power stages, video display drivers, and programmable power supplies to be simulated with IsSPICE. Shown below in Figure 10 are two other applications. The top schematic shows a motor control circuit with motor current a function of V_{in} . The bottom circuit is used for precision magnetic deflection.




Figure 10, Two circuits taken from the Apex Microtechnology data book illustrate the high power applications that can be simulated with IsSPICE and Apex's new device models. The models are available immediately on the *Intusoft Newsletter* floppy.